Relations between some characteristic lengths in a triangle

Wolfram Koepf and Markus Brede

Keywords: Pythagorean Theorem, Altitude Theorem, Cathetus Theorems, Groebner basis, polynomial elimination.

Abstract

The paper's aim is to note a remarkable (and apparently unknown) relation for right triangles, its generalization to arbitrary triangles and the possibility to derive these and some related relations by elimination using Groebner basis computations with a modern computer algebra system.

1 The Pythagorean group of theorems

We start by noting the Pythagorean group of theorems. Assume a right triangle T with vertices A, B and C in standard notation is given. Hence the right angle of T is at the vertex C, $a = \overline{BC}$ and $b = \overline{AC}$ denote the lengths of the two catheti and $c = \overline{AB}$ denotes the length of the hypotenuse.[1]



Figure 1: The standard notation in a right triangle

Furthermore, the length of the altitude CH with the hypotenuse as base is denoted by h. Finally by $q = \overline{AH}$ and $p = \overline{HB}$ we denote the lengths of the hypotenuse sections. Of course by construction the equation $c = p + q$ is valid.

The Pythagorean theorem is given in the whole triangle ACB as

$$a^2 + b^2 = c^2$$

as well as in the two smaller right angled triangles AHC and BHC as

[1] There seems to be no worldwide standard terminology. We will use this terminology throughout the paper.

$$p^2 + h^2 = a^2$$

and

$$q^2 + h^2 = b^2$$

respectively.

Furthermore, the following well-known identities are valid in the triangle *T*:

- (**Area Identity)** The double area of *T* can be computed as *ab* and as *ch*, hence *ab* = *ch*.
- (**Altitude Theorem)** The square of the altitude equals the product of the two hypotenuse sections $h^2 = pq$.
- (**Cathetus Theorems)** The square of a cathetus equals the product of its adjacent hypotenuse section and the hypotenuse; hence $a^2 = pc$ and $b^2 = qc$.

In Section 3, we will show that all these identities can be computed from our starting equations

$$\begin{aligned}
P_1 &= c - p - q = 0 \\
P_2 &= a^2 + b^2 - c^2 = 0 \\
P_3 &= p^2 + h^2 - a^2 = 0 \\
P_4 &= q^2 + h^2 - b^2 = 0
\end{aligned}$$

by elimination of variables.

2 A new identity

Next, we present the relation between the two catheti and the altitude of our right triangle *T*. This relation states that the sum of the *reciprocal* squares of the two catheti equals the reciprocal square of the altitude, i.e.

$$\frac{1}{a^2} + \frac{1}{b^2} = \frac{1}{h^2}. \tag{1}$$

This identity was given in [1].

The statement can easily be proven in different ways. Using the Pythagorean theorem and the area identity $a^2 b^2 = c^2 h^2$, we immediately get

$$\frac{1}{a^2} + \frac{1}{b^2} = \frac{a^2 + b^2}{a^2 \cdot b^2} = \frac{c^2}{c^2 \cdot h^2} = \frac{1}{h^2},$$

resp., alternatively

$$\frac{1}{a^2} + \frac{1}{b^2} = \frac{a^2 + b^2}{a^2 \cdot b^2} = \frac{c}{a^2} \cdot \frac{c}{b^2} = \frac{1}{p} \cdot \frac{1}{q} = \frac{1}{h^2},$$

where we have used the Pythagorean as well as the cathetus and the altitude theorem.

A third proof can be given as follows:

$$\frac{1}{a^2}+\frac{1}{b^2}=\frac{1}{pc}+\frac{1}{qc}=\frac{p+q}{pqc}=\frac{1}{pq}=\frac{1}{h^2}.$$

Finally, we can derive the assertion also by eliminating p, q and c in the relations

$$a^2+b^2=c^2, \tag{2}$$

$$a^2=pc,\;\; b^2=qc, \tag{3}$$

$$h^2=pq. \tag{4}$$

Starting with (4), substituting p and q by means of (3), we get

$$h^2=\frac{a^2}{c}\cdot\frac{b^2}{c}=\frac{a^2b^2}{c^2}.$$

Now, using (2) we get

$$h^2=\frac{a^2b^2}{a^2+b^2}=\frac{1}{\frac{1}{a^2}+\frac{1}{b^2}},$$

and this is again the above assertion.

3 Elimination of variables

In the previous section we showed how elimination of variables can be used to obtain new identities. The computation of Groebner bases with respect to lexicographical monomial orderings by Buchberger's algorithm makes elimination of variables in a polynomial system an algorithmic process, see e.g. [2]. For this purpose we use the Maple computer algebra system.

A Groebner basis of a set of polynomials consists of polynomials generating the same ideal. In particular, if the starting polynomials are zero, then all the Groebner basis polynomials are zero, too.

Let us start with the trivial statement

$$P_1=c-p-q=0$$

and the Pythagorean identities

$$P_2=a^2+b^2-c^2=0$$

$$P_3=p^2+h^2-a^2=0$$

$$P_4=q^2+h^2-b^2=0.$$

These are four polynomial equations in the six variables a, b, c, h, p, q. Hence we would like to eliminate any three of our variables to obtain a polynomial identity in the other three. This is exactly what Groebner basis computations with respect to lexicographical monomial orderings do.

We define our four polynomials in Maple[2]

```
> eqs:={c-p-q,a^2+b^2-c^2, p^2+h^2-a^2,q^2+h^2-b^2};
```

$$eqs := \{a^2+b^2-c^2, c-p-q, q^2+h^2-b^2, p^2+h^2-a^2\}$$

and load the Groebner basis package:

```
> with(Groebner):
```

To obtain the Altitude Theorem we have to eliminate the variables a, b and c. Hence, for the Groebner basis computation, we use a lexicographical ordering with a, b and c on top of the other variables:

```
> GB:=gbasis(eqs,plex(a,b,c,p,q,h));
```

$$GB := [-h^2+p\,q, c-p-q, -q^2-h^2+b^2, -p^2-h^2+a^2]$$

The computed Groebner basis of the ideal constituted by our four polynomials contains the polynomial

```
> op(1,GB);
```

$$-h^2+p\,q$$

The previous computation has shown that under our hypotheses the equation $-h^2+pq=0$ is valid, hence we have derived and proved the Altitude Theorem.

Similarly, to detect the first Cathetus Theorem, we eliminate the variables b, h and q:

```
> GB:=gbasis(eqs,plex(b,h,q,a,c,p));
```

$$GB := [a^2-c\,p, -c+p+q, h^2-c\,p+p^2, b^2-c^2+c\,p]$$

```
> op(1,GB);
```

$$a^2-c\,p$$

In a similar fashion, the second Cathetus Theorem is deduced.

Next, we generate the theorem of Section 2 again:

```
> GB:=gbasis(eqs,plex(c,p,q,a,b,h));
```

$$GB := [-b^2h^2+b^2a^2-h^2a^2, q^2+h^2-b^2, h^2q-a^2q+h^2p, b^2p-a^2q, -h^2+p\,q, p^2+h^2-a^2, c-p-q]$$

```
> op(1,GB);
```

$$-b^2h^2+b^2a^2-h^2a^2$$

Division by $a^2b^2h^2$ yields (1) again.

Eliminating other variables, the above method yields a polynomial identity in any three of the variables a, b, c, h, p, q.

Finally, how can we derive the Area Identity from our starting identities? Since the variables a, b, c and h should survive, we choose the lexicographical term order $p > q > h > a > b > c$ and get

```
> GB:=gbasis(eqs,plex(p,q,h,a,b,c));
```

$$GB := [a^2+b^2-c^2, b^4+c^2h^2-b^2c^2, -b^2+c\,q, c\,h^2-b^2c+q\,b^2, q^2+h^2-b^2, -c+p+q]$$

[2] For efficiency reasons Maple does not sort output alphabetically, but by memory allocation.

Of course, the first polynomial of the resulting Groebner basis is our starting one $a^2+b^2-c^2$, but the second polynomial contains also h:

```
> op(2,GB);
```

$$b^4+c^2h^2-b^2c^2$$

We can combine the corresponding identities properly to deduce the Area Identity. The following computation shows how this can be accomplished:

$$0=b^4+c^2h^2-c^2b^2=b^2\left(c^2-a^2\right)+c^2h^2-c^2b^2=c^2h^2-a^2b^2.$$

Note that although the polynomial $c^2h^2-a^2b^2$ is a member of the ideal considered, the polynomial $ch-ab$ is not:

```
> normalf(c^2*h^2-a^2*b^2,GB,plex(p,q,h,a,b,c));
```

$$0$$

```
> normalf(c*h-a*b,GB,plex(p,q,h,a,b,c));
```

$$c\,h-a\,b$$

From the point of view of ideal theory $ch-ab=0$ does not follow from our assumptions. However, since in a triangle all variables $a,b,c,h>0$, from $a^2b^2=c^2h^2$ the identity $a\,b=c\,h$ can be deduced.

4 Identities in a general triangle

The Altitude and Catheti Theorems are restricted to right angles. Therefore, it might be of interest how these theorems can be generalized to arbitrary triangles T. This is the goal of this section. Therefore we start with the following polynomials

$$P_1=c-p-q=0\,,$$

the Cosine Theorem in T

$$P_2=a^2+b^2-c^2-2\,a\,b\cos\gamma=0,$$

which constitiutes the generalized version of the Pythagorean theorem in a general triangle, and the Pythagorean identities

$$P_3=p^2+h^2-a^2=0$$
$$P_4=q^2+h^2-b^2=0.$$

Note that these polynomials depend on the angle γ at C.

To continue, we define therefore[3]

```
> eqs:={c-p-q,a^2+b^2-c^2-2*a*b*cosgamma,p^2+h^2-a^2,q^2+h^2-b^2};
```

[3] Note that the Groebner package cannot deal with the expression cos γ which we therefore replaced by the variable `cosgamma`.

$$eqs := \{ c - p - q,\ q^2 + h^2 - b^2,\ p^2 + h^2 - a^2,\ a^2 + b^2 - c^2 - 2\, a\, b\, cosgamma \}$$

Elimination of b, h and q yields a generalized first Cathetus Theorem

```
> GB:=gbasis(eqs,plex(b,h,q,a,c,p));
```

$$GB := [a^4\, cosgamma^2 - a^4 + cosgamma^2\, a^2\, c^2 + 2\, c\, p\, a^2 - 2\, c\, p\, a^2\, cosgamma^2 - p^2\, c^2,\\ -c + p + q,\ p^2 + h^2 - a^2,\ cosgamma^2\, a\, c^2 + cosgamma\, b\, p\, c + cosgamma^2\, a^3\\ - 2\, a\, cosgamma^2\, c\, p - a^3 + a\, c\, p,\ a\, b\, cosgamma - a^2 + c\, p,\ b^2 - a^2 - c^2 + 2\, c\, p]$$

given by the following polynomial

```
> res:=subs(cosgamma=cos(gamma),op(1,GB));
```

$$res := a^4 \cos(\gamma)^2 - a^4 + \cos(\gamma)^2\, a^2\, c^2 + 2\, c\, p\, a^2 - 2\, c\, p\, a^2 \cos(\gamma)^2 - p^2\, c^2$$

This is of course a much more difficult statement than the Cathetus Theorem. Nevertheless, if $\gamma = \pi/2$, then we get

```
> factor(eval(res,gamma=Pi/2));
```

$$-(-a^2 + c\, p)^2$$

by factorization. Note that we did not find any other rational multiple angle of π such that factorization, and therefore simplification, occurs in this statement.

Let us restate the new identity in the variables b, c, q and γ valid in a general triangle T:

$$a^2\left(a^2\cos^2\gamma - a^2 + \cos^2\gamma c^2 + 2pc - 2pc\cos^2\gamma\right) = c^2 p^2.$$

In this form one can easily see how this statement generalizes the Cathetus Theorem.

Our next computation concerns a generalization of the Altitude Theorem, derived by:

```
> GB:=gbasis(eqs,plex(a,b,c,p,q,h));
```

$$GB := [-p^2\, q^2 + 2\, q\, h^2\, p - h^4 + cosgamma^2\, h^4 + cosgamma^2\, q^2\, p^2 + cosgamma^2\, h^2\, p^2\\ + cosgamma^2\, q^2\, h^2,\ c - p - q,\ -q^2 - h^2 + b^2,\\ b\, q\, p + cosgamma\, a\, q^2 + cosgamma\, h^2\, a - b\, h^2,\ cosgamma\, h^2\, a\, p\\ + cosgamma\, a\, h^2\, q + b\, q\, p^2 - b\, q\, p^2\, cosgamma^2 - b\, h^2\, p - cosgamma^2\, h^2\, b\, q,\\ p\, q\, a - h^2\, a + cosgamma\, b\, p^2 + b\, cosgamma\, h^2,\ a\, b\, cosgamma + p\, q - h^2,\\ -p^2 - h^2 + a^2]$$

```
> res:=subs(cosgamma=cos(gamma),op(1,GB));
```

$$res := -p^2\, q^2 + 2\, q\, h^2\, p - h^4 + \cos(\gamma)^2\, h^4 + \cos(\gamma)^2\, q^2\, p^2 + \cos(\gamma)^2\, h^2\, p^2 + \cos(\gamma)^2\, q^2\, h^2$$

resulting in the statement

$$-p^2q^2 + 2qh^2p - h^4 + \cos^2\gamma h^4 + \cos^2\gamma q^2p^2 + \cos^2\gamma h^2p^2 + \cos^2\gamma q^2h^2 = 0.$$

Of course, for $\gamma = \pi/2$, we get the Altitude Theorem by factorization:

```
> factor(eval(res,gamma=Pi/2));
```

$$-(-h^2 + p\, q)^2$$

In this case, there is a second angle where factorization occurs, namely for $\gamma = \pi/4$ or 45 degrees. We get

```
> factor(eval(res,gamma=Pi/4));
```

$$-\frac{(-h^2 - q\,h - p\,h + p\,q)\,(-h^2 + q\,h + p\,h + p\,q)}{2}$$

We would like to determine which of the resulting relations

$$h^2 = pq \pm (ph + qh) = pq \pm ch \tag{5}$$

is valid. Since in a right triangle $h^2 = pq$, by lowering the angle at C and keeping h fixed, it is clear that both p and q must get smaller, hence the minus sign in (5) is impossible. Therefore we finally get for a triangle with angle $\pi/4$ at C the identity

$$h^2 = pq + ch$$

or

$$pq = h(h - c)$$

in particular

$$h - c \geq 0 \quad \text{or} \quad h \geq c.$$

Indeed, equality in this inequality occurs if the triangle is right-angled at A or B.

Finally, we deduce the generalization of (1) by elimination of c, p and q.

```
> GB: =gbasis(eqs,plex(c,q,p,a,b,h));
```

$$GB := [-b^2\,a^2 + h^2\,a^2 + b^2\,h^2 + a^2\,b^2\,cosgamma^2 - 2\,cosgamma\,b\,h^2\,a,\ p^2 + h^2 - a^2,$$
$$cosgamma^2\,b^2\,h^2\,q - h^4\,q + cosgamma\,b^3\,p\,a - cosgamma^3\,b^3\,p\,a$$
$$- cosgamma\,a\,p\,h^2\,b + cosgamma^2\,b^2\,h^2\,p + p\,b^2\,h^2 - h^4\,p,$$
$$h^2\,q\,a - cosgamma\,b\,h^2\,q - b^2\,p\,a + b^2\,p\,a\,cosgamma^2 + a\,p\,h^2 - cosgamma\,b\,h^2\,p,$$
$$b^2\,p - h^2\,p - h^2\,q + cosgamma\,a\,b\,q,\ -h^2\,p + cosgamma\,a\,b\,p + a^2\,q - h^2\,q,$$
$$a\,b\,cosgamma + p\,q - h^2,\ q^2 + h^2 - b^2,\ c - p - q\,]$$

```
> res:=subs(cosgamma=cos(gamma),op(1,GB));
```

$$res := -b^2\,a^2 + h^2\,a^2 + b^2\,h^2 + a^2\,b^2\cos(\gamma)^2 - 2\cos(\gamma)\,b\,h^2\,a$$

After division by $a^2b^2h^2$ and replacing $1-\cos^2\gamma$ by $\sin^2\gamma$, this results in

$$\frac{\sin^2\gamma}{h^2} = \frac{1}{a^2} + \frac{1}{b^2} - \frac{2\cos\gamma}{ab},$$

which obviously generalizes (1) in a nice way.

References

[1] Brede, Markus: Eine Bemerkung zur Satzgruppe des Pythagoras. Preprint 19/04, Department of Mathematics, University of Kassel, 2004.

[2] Cox, D.; Little, J.; O'Shea, D.: Ideals, Varieties and Algorithms. Springer, New York, second edition, 1997.